
1. [bookmark: _Toc122434485][bookmark: _Toc66434900]Overview
This document lists all the essential changes needed to correct problems in the TTCN implementation of test case 11.3.9 which is part of the 5G NR5GC test suite. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence. 
Contact:	Shaun Harry
	shaun.harry@keysight.com
0. [bookmark: _Toc66434901]Verification Test Summary 
Test Case: 	11.3.9.NR5GC
ATS Version:	iwd-TTCN3-B2020-09_D20wk50
System Simulator used:	Keysight S8704A Protocol Conformance Toolset
UE used:	Qualcomm SM8350 + SDX60
Verification Status:	PASS

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc66434902]Corrections required
2.1 [bookmark: _Toc66434903]Change 1
	Function name
	f_TC_11_3_9_NR5GC()

	Reason for change
	1. SIB1 does not contain UAC barring information as mentioned in the Table 11.3.9.3.3-2, because the call to f_NR_CellInfo_SetSIB1_UacBarringInfo() is done after the cell is configured

2. DNN criteria components in Operator Defined Access Categories should be encoded as : 

(24.501 9.11.3.38) :  DNN criteria type (00) + DNN length-value pair count (01) + DNN value length + DNN value encoded according 23.003.

(23.003 9.1) : "The APN consists of one or more labels. Each label is coded as a one octet length field followed by that number of octets coded as 8 bit ASCII characters"

e.g. For DNN label “ABCD” the encoding should be :

00 – criteria type = DNN
01 – count of DNN length-value pairs 
05 – DNN length 
04 41 42 43 44 – DNN (ABCD)

3) Following completion of test body, UE will be registered on cell 12.


	Summary of change
	1. Move setting of SIB1 contents for UAC Barring info to before cell configuration

2. Modify const_Criteria_DNN to “00”. When building the criteria field, include DNN count (01) and the 2 required length fields.

3. Use cell 12 for postamble

	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	




Before Change:
	function f_TC_11_3_9_NR5GC() runs on NR5GC_PTC
  { //UAC / Access Identity 0 / ODAC / PLMN / RPLMN / not EPLMN
    const charstring const_DNNValue := "ABCD";
    const octetstring const_Criteria_DNN := '01'O;  //DNN Type
    var template (value) NR_CellPowerList_Type v_CellPowerList_T0;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var octetstring v_OctetString;
    var octetstring v_Criteria;
 
  
    v_OctetString := char2oct(const_DNNValue);
    
    v_Criteria := const_Criteria_DNN & v_OctetString;
    
    // System information combination NR-1 as defined in TS 38.508-1 clause 4.4.3.1.2 is used in NR cells.
    f_NR5GC_Init(NR_1);

    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    f_NR_CellConfig_Def(nr_Cell12);
    
     // Table 11.3.9.3.3-2: SIB1 for NR Cell 1 and NR Cell 12
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1,
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(33,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B)));
    
    v_CellPowerList_T0 := {//@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2),
      cs_NR_CellPower (nr_Cell12, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T0);

    // The UE is registered on PLMN1 (NR Cell 1) and in state 3N-A on NR Cell 1(serving cell) by using the procedure described in TS 38.508-1 [4] clause 4.5.2.2 with 'connected without release'
    // except that the REGISTRATION ACCEPT message indicates Operator Defined Access Category ""33"" as described in Table 11.3.9.3.3-1.
    v_GMM_MobilityInfo := f_NR5GC_Preamble_Steps1_13(nr_Cell1);
    
    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,
                                                   v_GMM_MobilityInfo,  // @sic R5-206426 sic@
                                                   f_NR5GC_GetMobileIdGUTI (nr_Cell1),
                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell1)), {bit2oct(f_NR_CellInfo_GetTAC (nr_Cell1))}),
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   cs_OperatorAccessCatDefinitions(cs_AccessCatDefinition ('00'O,'1'B,'00001'B, v_Criteria)));
    
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell1,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
                                   
    // Steps 15-19a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed.
    f_NR5GC_RRC_Idle_Steps15_20(nr_Cell1,TESTMode_OFF);
    
    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1);

    f_NR_TestBody_Set(true);

    f_NR_TestBody_Set(true);
    fl_TC_11_3_9_TestBody();
    f_NR_TestBody_Set(false);
    
    f_NR_Postamble(nr_Cell1, STATE_CONNECTED_3A);
  }

	




After Change:
	function f_TC_11_3_9_NR5GC() runs on NR5GC_PTC
  { //UAC / Access Identity 0 / ODAC / PLMN / RPLMN / not EPLMN
    const charstring const_DNNValue := "ABCD";
    const octetstring const_Criteria_DNN := '00'O;  //DNN Type
    var template (value) NR_CellPowerList_Type v_CellPowerList_T0;
    var template (value) NG_NAS_DL_Message_Type v_MsgToSend;
    var GMM_MobilityInfo_Type v_GMM_MobilityInfo;

    var octetstring v_OctetString;
    var octetstring v_Criteria;
   
 
   v_OctetString := int2oct(lengthof(const_DNNValue),1) & char2oct(const_DNNValue); 
    
    v_Criteria := const_Criteria_DNN & ‘01’O & int2oct(lengthof(v_OctetString),1) & v_OctetString;
    
    // System information combination NR-1 as defined in TS 38.508-1 clause 4.4.3.1.2 is used in NR cells.
    f_NR5GC_Init(NR_1);
    
    // Table 11.3.9.3.3-2: SIB1 for NR Cell 1 and NR Cell 12
    f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1,
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(33,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B)));

    //Create and configure cell
    f_NR_CellConfig_Def(nr_Cell1);
    f_NR_CellConfig_Def(nr_Cell12);

     // Table 11.3.9.3.3-2: SIB1 for NR Cell 1 and NR Cell 12
    /* MOVED f_NR_CellInfo_SetSIB1_UacBarringInfo(nr_Cell1,
                                         cs_SIB1_UacBarringInfo(cs_UAC_BarringPerCatList_1Entry(33,1), -, cs_UAC_BarringInfoSetList_1Entry(p00, s16, '1111111'B))); */

    
  
    v_CellPowerList_T0 := {//@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2),
      cs_NR_CellPower (nr_Cell12, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T0);

    // The UE is registered on PLMN1 (NR Cell 1) and in state 3N-A on NR Cell 1(serving cell) by using the procedure described in TS 38.508-1 [4] clause 4.5.2.2 with 'connected without release'
    // except that the REGISTRATION ACCEPT message indicates Operator Defined Access Category ""33"" as described in Table 11.3.9.3.3-1.
    v_GMM_MobilityInfo := f_NR5GC_Preamble_Steps1_13(nr_Cell1);
    
    v_MsgToSend := f_Get_NG_RegistrationAcceptMsg (Initial_Secure,
                                                   v_GMM_MobilityInfo,  // @sic R5-206426 sic@
                                                   f_NR5GC_GetMobileIdGUTI (nr_Cell1),
                                                   cs_NG_TAIListNonConsecutive(f_NR_Asn2Nas_PlmnId(f_NR_CellInfo_GetPLMN (nr_Cell1)), {bit2oct(f_NR_CellInfo_GetTAC (nr_Cell1))}),
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   -,-,-,-,-,
                                                   cs_OperatorAccessCatDefinitions(cs_AccessCatDefinition ('00'O,'1'B,'00001'B, v_Criteria)));
    
    SRB.send(cas_NR_SRB_NasPdu_REQ(nr_Cell1,
                                   tsc_NR_RbId_SRB1, // No SRB2 so must be on SRB1
                                   -,
                                   cs_NG_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_MsgToSend)));
                                   
    // Steps 15-19a1 of Table 4.5.2.2-2 of the generic procedure in TS 38.508-1 [4] are performed.
    f_NR5GC_RRC_Idle_Steps15_20(nr_Cell1,TESTMode_OFF);
    
    f_NR5GC_RRC_ConnectedState3N_Def(nr_Cell1);

    f_NR_TestBody_Set(true);

    f_NR_TestBody_Set(true);
    fl_TC_11_3_9_TestBody();
    f_NR_TestBody_Set(false);
    
    f_NR_Postamble(nr_Cell12, STATE_CONNECTED_3A);
  }




[bookmark: _Toc53583567]
2.2 [bookmark: _Toc66434904]Change 2
	Function name
	function fl_TC_11_3_9_TestBody()

	Reason for change
	1. The implementation of step 4 has the issue that if a NAS message is received but the call to f_Check_NG_PDUSessionEstablishmentReq() returns false (e.g. a different NAS message, or a PDU SESSION ESTABLISHMENT REQUEST with non-matching DNN is received) then the execution will continue to step 5 as the unexpected NAS message has been consumed and so will not trigger the default handler.

2. At step 9 , the SERVICE REQUEST from the UE will have service type 0000 (“signalling”) currently expected value is 0010 (“mobile terminated services”). 



	Summary of change
	1. Modify the checking at step 4 so that any unexpected NAS message will cause the test case to fail at this step.

2. Correct service type in f_Check_NG_ServiceReqMsg()


	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	




Before Chang
	function fl_TC_11_3_9_TestBody() runs on NR5GC_PTC
  {
    const charstring const_DNNValue := "ABCD";
    
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_RadioBearerId_Type v_SRB:= tsc_NR_RbId_SRB2;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo_Type;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template DNN v_DNN;
    var boolean v_ReceiveRRCSetupRequest;


    timer t_Wait := 20.0;
    
    
    v_DNN := cs_DNN(-, f_DomainName_Encode(const_DNNValue));//R5s210148
    
    //@siclog "Steps 3" siclog@
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, const_DNNValue);
    
   //@siclog "Steps 4" siclog@
   // Check: Does the UE transmit a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is the same value in UL NAS TRANSPORT message in preamble within 20s?
    t_Wait.start;
    alt{
        []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
            {
               if (f_Check_NG_PDUSessionEstablishmentReq(v_GSM_MobilityInfo_Type,
                                                  v_ReceivedAsp.Signalling.Nas[0].Pdu, -, -, -, -, -, -, -, -, -, -, -, -, -, v_DNN))
                {
                    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
                }
               
               
            }
        []t_Wait.timeout
            {
               f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
            }
    }
    // @siclog "Step 5" siclog@
    //The SS adjusts the NR Cells power levels according to row "T1" in table 11.3.9.3.2-1/2.
     v_CellPowerList_T1 := { //@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell12, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    
     // @siclog "Step 6" siclog@
    //The mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 is performed on NR Cell 12 and release RRC connection.
    f_NR5GC_MobilityRegistration(nr_Cell12);
    
    //@siclog "Steps 7" siclog@
    //Make the UE attempt to establish a PDU session for the DNN defined in Table 11.3.9.3.3-1. This can be done by an AT/MMI command.
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, const_DNNValue);
    
    //@siclog "Step 8" siclog@
    //Check: Does the UE transmit a RRCSetupRequest message and establishmentCause is set to mo-Data within 20s?
    t_Wait.start;
    v_ReceiveRRCSetupRequest := f_NR_RRC_SetupRequest_Common( nr_Cell12, t_Wait, cr_NR_EstablishmentCause_MoData);
    if (v_ReceiveRRCSetupRequest) {
         f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
    } else {
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");
    }

    //@siclog "Steps 9 - 14" siclog@
    //Steps 3-8 of the generic procedure for NR RRC_Connected specified in TS 38.508-1 Table 4.5.4.2-3 are performed.
    f_NR_RRCSetup_Def (nr_Cell12);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
                                          tsc_SHT_IntegrityProtected);
                                          
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)),'0010'B)) {  // @sic R5s200654 subject to prose CR approval sic@
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg);

    //@siclog "Steps 15 - 16" siclog@
    //The UE transmits a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is set to the same value with the DNN value of Operator-defined access category definitions defined in Table 11.3.9.3.3-1
    //The SS transmits PDU SESSION ESTABLISHMENT ACCEPT message.
    f_NR5GC_PDUSessionEstablishment (nr_Cell12, 1, -, -, - , v_DNN, -, v_DNN);//R5s210148
    
  }

	




After Change:
	function fl_TC_11_3_9_TestBody() runs on NR5GC_PTC
  {
    const charstring const_DNNValue := "ABCD";
    
    var NG_NAS_MSG_Indication_Type v_ReceivedMsg;
    var NR_RadioBearerId_Type v_SRB:= tsc_NR_RbId_SRB2;
    var GSM_MobilityInfo_Type v_GSM_MobilityInfo_Type;
    var NR_SRB_COMMON_IND v_ReceivedAsp;
    var template (value) NR_CellPowerList_Type v_CellPowerList_T1;
    var template DNN v_DNN;
    var boolean v_ReceiveRRCSetupRequest;


    timer t_Wait := 20.0;
    
    
      v_DNN := cs_DNN(-, f_DomainName_Encode(const_DNNValue));//R5s210148
    
    //@siclog "Steps 3" siclog@
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, const_DNNValue);
    
   //@siclog "Steps 4" siclog@
   // Check: Does the UE transmit a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is the same value in UL NAS TRANSPORT message in preamble within 20s?
    t_Wait.start;
    alt{
        []SRB.receive(car_NR_SRB_NasPdu_IND(nr_Cell1, v_SRB, cr_NG_NAS_IndWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered))) -> value v_ReceivedAsp
            {
               /* REMOVED if (f_Check_NG_PDUSessionEstablishmentReq(v_GSM_MobilityInfo_Type,
                                                  v_ReceivedAsp.Signalling.Nas[0].Pdu, -, -, -, -, -, -, -, -, -, -, -, -, -, v_DNN))
                {*/
                    f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
                //REMOVED }      
            }
      []t_Wait.timeout
      {
         f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 4");
      }
    }
    // @siclog "Step 5" siclog@
    //The SS adjusts the NR Cells power levels according to row "T1" in table 11.3.9.3.2-1/2.
     v_CellPowerList_T1 := { //@sic R5-205388 sic@
      cs_NR_CellPower (nr_Cell1, tsc_NR_NonSuitableOffCellSSS_EPRE, tsc_NR_NonSuitableOffCellSSS_EPRE),
      cs_NR_CellPower (nr_Cell12, tsc_NR_ServingCellSSS_EPRE_FR1, tsc_NR_ServingCellSSS_EPRE_FR2)
    };
    
    f_NR_SetCellPowerList(v_CellPowerList_T1);
    
     // @siclog "Step 6" siclog@
    //The mobility registration updating procedure described in TS 38.508-1 [4] Table 4.9.5.2.2-1 is performed on NR Cell 12 and release RRC connection.
    f_NR5GC_MobilityRegistration(nr_Cell12);
    
    //@siclog "Steps 7" siclog@
    //Make the UE attempt to establish a PDU session for the DNN defined in Table 11.3.9.3.3-1. This can be done by an AT/MMI command.
    f_UT_RequestPDUSession(UT, f_NR5GC_MobileInfo_GetNoOfSessions()+1, const_DNNValue);
    
    //@siclog "Step 8" siclog@
    //Check: Does the UE transmit a RRCSetupRequest message and establishmentCause is set to mo-Data within 20s?
    t_Wait.start;
    v_ReceiveRRCSetupRequest := f_NR_RRC_SetupRequest_Common( nr_Cell12, t_Wait, cr_NR_EstablishmentCause_MoData);
    if (v_ReceiveRRCSetupRequest) {
         f_NR_PreliminaryPass(__FILE__, __LINE__, "Step 8");
    } else {
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 8");
    }

    //@siclog "Steps 9 - 14" siclog@
    //Steps 3-8 of the generic procedure for NR RRC_Connected specified in TS 38.508-1 Table 4.5.4.2-3 are performed.
    f_NR_RRCSetup_Def (nr_Cell12);
    
    v_ReceivedMsg := f_NR_RRCSetupComplete_Def(nr_Cell12,
                                          cr_38508_RRCSetupComplete(tsc_NR_RRC_TI_Def, ?,- , *),
                                          tsc_SHT_IntegrityProtected);
                                          
    if (not f_Check_NG_ServiceReqMsg (v_ReceivedMsg.Pdu, cs_NAS_KeySetIdentifier_lv(f_NR_SecurityKSIamf_Get (), tsc_NasKsi_NativeSecurityContext), f_NG_S_TMSI2MobileIdentity(omit, f_NR5GC_CellInfo_GetGutiParameters (nr_Cell1)),'0000'B)) {  // @sic R5s200654 subject to prose CR approval sic@
         f_NR_SetVerdictFailOrInconc(__FILE__, __LINE__, "Service Request Message Failed");
    }
    
    f_NR5GC_RRC_ConnectedState3N_Steps5_8(nr_Cell12, v_ReceivedMsg);

    //@siclog "Steps 15 - 16" siclog@
    //The UE transmits a PDU SESSION ESTABLISHMENT REQUEST and the DNN in UL NAS TRANSPORT message is set to the same value with the DNN value of Operator-defined access category definitions defined in Table 11.3.9.3.3-1
    //The SS transmits PDU SESSION ESTABLISHMENT ACCEPT message.
    f_NR5GC_PDUSessionEstablishment (nr_Cell12, 1, -, -, - , v_DNN, -, v_DNN);//R5s210148
    
  }





2.3 [bookmark: _Toc66434905]Change 3
	Function name
	template OperatorAccessCatDefinitions

	Reason for change
	The IE length of “Operator-defined access category definitions" is incorrectly calculated. The current implementation is calling encvalue directly to encode p_AccessCatDefinition which actually returns a bitstring. 

	Summary of change
	Convert the result of encvalue() to an octetstring before taking the length.

	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	




Before Change:
	template (value) OperatorAccessCatDefinitions cs_OperatorAccessCatDefinitions (template(value)AccessCatDefinition p_AccessCatDefinition):=
  {
    iei := '76'O,       /* present in case of TLV; omit in case of LV */
    iel := int2oct(lengthof(encvalue(p_AccessCatDefinition)), 2),
    definitions := {p_AccessCatDefinition}
  };

	




After Change:
	template (value) OperatorAccessCatDefinitions cs_OperatorAccessCatDefinitions (template(value)AccessCatDefinition p_AccessCatDefinition):=
  {
    iei := '76'O,       /* present in case of TLV; omit in case of LV */
    iel := int2oct(lengthof(bit2oct(encvalue(p_AccessCatDefinition))), 2), 
    definitions := {p_AccessCatDefinition}
  };



2.4 [bookmark: _Toc66434906]Change 4
	Function name
	template (value) AccessCatDefinition

	Reason for change
	The length for an individual access category definition is incorrectly calculated. Following 24.501 9.11.3.38 there should be 4 byte of overheaded added to the lemgth of “p_Criteria” (for precedence, PSAC/spare/category number, criteria length and spare/stanrdised access category)

	Summary of change
	Correct length calculation

	TTCN module
	UAC_NR5GC.ttcn

	MCC160 Comment
	





Before Change:
	template (value) AccessCatDefinition cs_AccessCatDefinition (O1_Type p_Precedence, B1_Type p_Psac, B5_Type p_CatNum, octetstring  p_Criteria):=
   {
    iel:= int2oct(lengthof(p_Criteria)+3, 1),
    precendence := p_Precedence,
    psac := p_Psac,
    spare2 := '00'B,
    catNum := p_CatNum,
    criteriaLength :=int2oct(lengthof(p_Criteria), 1),
    criteria := p_Criteria,
    spare3 := '000'B,
    standardAccessCat := '00111'B    //Access category 7
 };




After Change:
	template (value) AccessCatDefinition cs_AccessCatDefinition (O1_Type p_Precedence, B1_Type p_Psac, B5_Type p_CatNum, octetstring  p_Criteria):=
   {
    iel:= int2oct(lengthof(p_Criteria)+4, 1),
    precendence := p_Precedence,
    psac := p_Psac,
    spare2 := '00'B,
    catNum := p_CatNum,
    criteriaLength :=int2oct(lengthof(p_Criteria), 1),
    criteria := p_Criteria,
    spare3 := '000'B,
    standardAccessCat := '00111'B    //Access category 7
  };





1. [bookmark: _Toc122434493][bookmark: _Toc295288970][bookmark: _Toc325725665][bookmark: _Toc66434907]Branches executed
[bookmark: _Toc122434494][bookmark: _Toc295288971][bookmark: _Toc325725666]The test case was executed on NR band n78
3 [bookmark: _Toc66434908]Execution Log Files
3.1 [bookmark: _Toc66434909]Qualcomm SM8350 + SDX60
The Qualcomm SM8350+SDX60 UE passed this test case on Keysight S8704A Protocol Conformance Toolset. The documentation below is enclosed as evidence of the successful test case run [1]:
· Test case execution log file:
TC_11_3_9_LOG.html
· PICS/PIXIT parameter file:
      TC_11_3_9_PIXIT.xml



4 [bookmark: _Toc122434496][bookmark: _Toc295288973][bookmark: _Toc325725668][bookmark: _Toc66434910]References 
	[1]
	R5s210308 : NR5GC FR1 : Supporting information for addition of NR5GC multilayer test case 11.3.9





